76-3

TM-…-3—CALFLO: A BASIC Language Program Used in the Calibration of Vol-O-Flo Flow Meters—by Maurice F. Funke

# CALFLO: A BASIC Language Program Used in the Calibration of Vol-O-Flo Flow Meters

March 1976

U.S. Army Materiel Command
HARRY DIAMOND LABORATORIES
Adelphi, Maryland 20783

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

# REPORT DOCUMENTATION PAGE

READ INSTRUCTIONS BEFORE COMPLETING FORM

| | | |
|---|---|---|
| 1. REPORT NUMBER<br>HDL-TM-76-3 | 2. GOVT ACCESSION NO. | 3. RECIPIENT'S CATALOG NUMBER |
| 4. TITLE (and Subtitle)<br>CALFLO: A BASIC Language Program Used in the Calibration of Vol-O-Flo Flow Meters | | 5. TYPE OF REPORT & PERIOD COVERED<br>Technical Memorandum<br>6. PERFORMING ORG. REPORT NUMBER |
| 7. AUTHOR(s)<br>Maurice F. Funke | | 8. CONTRACT OR GRANT NUMBER(s)<br>DA: 1T162114AH73 |
| 9. PERFORMING ORGANIZATION NAME AND ADDRESS<br>Harry Diamond Laboratories<br>2800 Powder Mill Road<br>Adelphi, MD 20783 | | 10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS<br>Program: 6.21.14.A |
| 11. CONTROLLING OFFICE NAME AND ADDRESS<br>US Army Materiel Development & Readiness Command<br>Alexandria, VA 22333 | | 12. REPORT DATE<br>March 1976<br>13. NUMBER OF PAGES<br>23 |
| 14. MONITORING AGENCY NAME & ADDRESS(if different from Controlling Office) | | 15. SECURITY CLASS. (of this report)<br>Unclassified<br>15a. DECLASSIFICATION/DOWNGRADING SCHEDULE |

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

HDL Project: 304634
DRCMS Code: 612116.11.H7300

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Calibration
Flow meter
Fluidics

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

This report describes CALFLO, a Wang 2200 BASIC program that, with data obtained from the flow meter calibration equipment located in HDL's calibration branch, presents a printout of the data, coefficients of the calibration equations, an analysis of data error, and plots of both normalized data and calibration curves for flow in either direction through a Vol-O-Flo meter. A derivation of the equations used for this calibration also is presented.

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE



UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

## CONTENTS



## FIGURES



## TABLE

## 1. INTRODUCTION

There is a program at the Harry Diamond Laboratories to begin a more intense testing and analysis of fluidic component characteristics. Because of the magnitude of the effort required to obtain precise data on fluidic elements, an effort has been made to standardize and automate as much of this procedure as possible. This effort has resulted in the construction of a fluidic test facility that includes an automated data acquisition system and computer control of test conditions. A computer program titled "Fluid" has been written to perform quasi-static performance data reduction and, specifically, to provide the information needed to make up the digital component fact sheet adopted by the Government Fluidics Coordination Group. In addition, a standard interconnection pattern for certain in-house components has been advanced.

This report describes CALFLO, a program written for the Wang 2200B programmable calculator, which can be used to obtain calibration information on Vol-O-Flo meters. The input to the program is the data collected with the HDL flow meter calibration equipment. The program output includes a printout of the data, the coefficients of the second order, least-squares curve fits of the corrected data for both flow directions, an analysis of data error, and plots of both normalized data and calibration curves.

## 2. DERIVATION OF EQUATIONS

The Vol-O-Flo meter is a transducer that provides a differential pressure signal ($\Delta P$) that is a function of the volume flow passing through its midpoint. The governing equation for this transducer[1] is given by

$$\Delta P = b\mu_m Q_m + c\rho_m Q_m^2 \quad , \tag{1}$$

where

- b,c are constants,
- $\mu$ is kinematic viscosity,
- Q is volume flow,
- $\rho$ is density,
- m refers to conditions at the midpoint of the meter.

[1] *D. I. Steele and L. B. Orbach, Theory and Use of the Vol-O-Flow Meter, National Instrument Laboratories, Inc. (January 1963).*

The first term on the right-hand side of equation (1) is due to Poiseuille flow. The second term is due to orifice effects. For laminar flow, the first term is much larger than the second.

Assuming steady flow, the continuity equation for the mass flow through the meter and the calibration system is

$$\rho_m Q_m = \rho Q = M \quad , \tag{2}$$

where M is mass flow rate.

A third equation is the ideal gas law,

$$\rho T/p = \text{constant}, \tag{3}$$

where T is temperature and p is pressure.

When a Vol-O-Flo meter is calibrated, $Q_m$, the volume flow at the meter midpoint, is not measured directly. Instead, a known volume of gas is collected at a constant flow rate over a measured period of time. However, it is possible to calculate $Q_m$ with equations (2) and (3), if the appropriate pressures and temperatures also are known. This calculation is as follows:

$$Q_v = V/t, \tag{4}$$

where

V is the known volume,

t is the measured time,

v refers to conditions at the known volume.

Equation (2) gives

$$Q_m = \rho_v Q_v / \rho_m \quad . \tag{5}$$

Equation (3) gives

$$\frac{\rho_v}{\rho_m} = \frac{p_v T_m}{p_m T_v} \quad . \tag{6}$$

Assuming that $T_m = T_v$, then combining equations (5) and (6) gives

$$Q_m = \frac{p_v}{p_m} Q_v \quad . \tag{7}$$

From equation (1), it is seen that for a given value of $Q_m$, $\Delta P$ is a function of temperature and pressure, since $\mu$ and $\rho$ are temperature and pressure dependent. Since the $(C\rho Q^2)$ term is very small, the changes in $\rho$ can be neglected over moderate temperature and pressure ranges. This approach allows for the absorption of the temperature and pressure corrections of $\mu$ by the $\Delta P$ term.

To correct the measured value of $\Delta P$ for the temperature and pressure variations of $\mu$, the following derivation is used. Since $\Delta P$ is a linear function of $\mu$,

$$\Delta P = \Delta P\left(\mu_s + \Delta\mu\right) = \Delta P\left(\mu_s\right) + \frac{\partial(\Delta P(\mu))}{\partial\mu} \Delta\mu \ , \tag{8}$$

where

$\Delta P\left(\mu_s + \Delta\mu\right)$ is the measured value of $\Delta P$ at some value of $\mu = \mu_s + \Delta\mu$,

$\mu_s$ is the value of $\mu$ at standard conditions of temperature (20°C) and pressure (1 bar),

$\Delta\mu$ is the difference between $\mu_s$ and the actual value of $\mu$.

From this discussion and equation (1),

$$\frac{\partial(\Delta P)}{\partial\mu} = bQ_m \simeq \frac{\Delta P\left(\mu_s\right)}{\mu_s} \quad . \tag{9}$$

Assuming that any deviations of the value of $\mu$ from a standard value can be approximated by a linear function and using the chain rule, one obtains

$$\Delta\mu = \frac{\partial\mu}{\partial T}\,\Delta T + \frac{\partial\mu}{\partial p}\,\Delta p \quad , \tag{10}$$

where

$\Delta T$ is $T_s - T_m$,

$\Delta p$ is $p_s - p_m$,

s refers to standard conditions (20°C, 1 bar).

Combining equations (8) to (10), one obtains

$$\Delta P = \Delta P_s + \frac{\Delta P_s}{\mu_s}\left(\frac{\partial\mu}{\partial T}\,\Delta T + \frac{\partial\mu}{\partial p}\,\Delta p\right) \quad , \tag{11}$$

where

$$\Delta P_s = \Delta P\left(\mu_s\right) \quad .$$

This can be rearranged to give

$$\Delta P_s = \frac{\Delta P}{1 + \frac{1}{\mu_s}\frac{\partial\mu}{\partial T}\,\Delta T + \frac{1}{\mu_s}\frac{\partial\mu}{\partial p}\,\Delta p} \quad . \tag{11a}$$

The values of $\frac{1}{\mu_s}\frac{\partial\mu}{\partial T}\,\Delta T$ and $\frac{1}{\mu_s}\frac{\partial\mu}{\partial p}\,\Delta p$ for various gasses are given in table I.

TABLE I. SPECIFIC GRAVITY AND VISCOSITY AND THEIR CHANGES WITH TEMPERATURE AND PRESSURE FOR VARIOUS GASSES

| Gas | Specific gravity (Air = 1) | $\mu_s$ (N-s/m²) x $10^7$ at $T_m$ = 20°C, $p_m$ = 1 bar | % Change in $\mu$ for change in $T_m$ of 1°C | % Change in $\mu$ for change in $p_m$ of 1 atm |
|---|---|---|---|---|
| Air | 1.000 | 181.5 | 0.263 | 0.13 |
| Argon | 0.5963 | 225.2 | 0.274 | 0.14 |
| Carbon dioxide | 1.5290 | 146.5 | 0.311 | 0.25 |
| Helium | 0.13804 | 195.4 | 0.227 | 0.011 |
| Hydrogen | 0.06952 | 88.22 | 0.230 | 0.016 |
| Nitrogen | 0.96724 | 175.4 | 0.256 | 0.13 |
| Oxygen | 1.10527 | 202.7 | 0.259 | 0.13 |

Equation (1) can be rewritten as

$$\Delta P_s = BQ_m + CQ_m^2 \quad , \tag{12}$$

where

$B = b\mu_s$ ,

$C = c\rho_s$ ,

The calibration equation now becomes

$$\Delta P_s = A + BQ_m + CQ_m^2 \quad , \tag{12a}$$

where A is the zero offset of a least-squares curve fit to the data. This can be written also as

$$Q_m = \frac{-B + \left(B^2 - 4C\left(A - \Delta P_s\right)\right)^{1/2}}{2C} \quad . \tag{12b}$$

If the gas at the midpoint of the flow meter is not at standard conditions (20°C, 1 bar), then the value of $\Delta P_s$ is not measured at the signal ports of the flow meter. However, equation (11a) can be used to obtain $\Delta P_s$ from the measured value.

The value of $\Delta P_s$ can be used to obtain the volume flow rate at the meter midpoint conditions, $Q_m$, by visual use of the plot of the calibration curve fit or by use of the value of $\Delta P_s$ in equation (12b).

The volume flow rate that would exist at any arbitrary conditions of temperature and pressure can be obtained by combining equations (5) and (6):

$$Q_i = \frac{p_m T_i}{p_i T_m} Q_m , \qquad (13)$$

where i refers to any arbitrary conditions of T and p.

Equation (12a) is used to obtain the least-squares curve fits for the corrected calibration data on the flow meter.

## 3. PROGRAM INPUT AND OUTPUT

This program was written in BASIC for a Wang 2200B calculator with 12k core. The peripherals used include a keyboard, cathode ray tube (CRT) display, 80-column thermal printer, and analog flatbed plotter 10 x 15 in. The plotted output requires graph paper 10 by 15 in. with the plotter zero set at the lower left corner of the grid and the plotter scale set 8 in. to the right and 8 in. above that point. Appendix A lists the program.

The data used by this program are entered from the keyboard. The user is prompted to enter the data by the CRT screen. The units in which the data should be entered (included in the prompt statements) are meant to reflect not any standard unit system, but rather the units in which the instrumentation used to calibrate the flow meters is scaled (fig. 1).

To provide a header label for program output, the program asks for specification of the date, meter serial number, and meter range. The meter range value is used also to determine the scale value of the plot. Since the Vol-O-Flo meter is usually fairly linear up to twice its nominal flow range, useful data can be obtained for flow levels up to twice the nominal range. Therefore, the plots of the calibration curve fits are automatically scaled to twice the nominal range of the meter.

The calibration data are entered a point at a time in the following order:

(a) Left manometer reading (cmwc)

(b) Right manometer reading (cmwc)

(c) Back pressure (kPa)

(d) Volume (liters)

(e) Time (min)

(f) Temperature (°C)

(g) Volume pressure (inwc)

(h) Barometric pressure (mm Hg)




Figure 1. Schematic of HDL flow meter calibration equipment.

Before any data are entered from the keyboard, a data point is automatically generated by the program. This point corresponds to the condition $\Delta P = Q = 0$. As the values for each complete data point are entered, they are printed out on the thermal printer. This printing allows the user to check his input data before program calculations begin.

If an incorrect value is entered, a correction can be made after all the data for the current flow direction are entered. The program recognizes that the last datum has been entered when the value for the left manometer reading is entered as 9.999. At that point, no additional data values can be entered, but those already entered can be changed.

Changes are made in the following manner: The CRT provides a prompt, asking if changes are to be made. If so, it prompts again, asking for the data point number, which was printed out with the data on the printer. (Its value is one higher than the order in which the data were entered, since the program generated the first point for $\Delta P = Q = 0$.) The program then asks for all the values for that data point, as listed above, and prints them out for subsequent verification. The CRT again provides a prompt, asking if changes are to be made. This sequence continues until all changes have been made.

At this point, the program calculates a second-order least-squares curve fit to the data. It then prints a data analysis table comparing the individual data point to that curve fit. The table includes the following parameters:

(a) $\Delta P_s$

(b) $Q_m$ (calculated from data)

(c) $Q_m$(calculated from curve fit)

(d) Difference between $Q_m$ from data and $Q_m$ from curve fit, (b) - (c)

(e) Percent ratio of difference between $Q_m$ from data and $Q_m$ from curve fit to $Q_m$ from data, $[(d)/(b)] \cdot 100$

(f) Percent ratio of difference between $Q_m$ from data and $Q_m$ from curve fit to full range flow of flow meter, $[(d)/R] \cdot 100$, where R = range of flow meter.

The program next prints out a value for the sum of the squares of the deviation, based on (d). At this point, the user can look at the data analysis table (column f) and see if any data points diverge significantly from the curve fit. If there are such points, the user should check that the correct data were entered. An entry error can be corrected, since the program again asks for data corrections at this point. Since an error could have been made in taking the data, the program allows the deletion of up to three data points, once all desired data corrections have been made. Once these corrections and deletions are made, a new curve fit is calculated, and a second data analysis table is printed. The program continues to cycle through this correction, deletion, curve fit, and data analysis routine, until the user makes no corrections or deletions. When this routine is completed, a table is printed of $\Delta P$ versus Q for nominal values of $\Delta P$ and values of Q calculated from the curve fit. The program also provides plots for both 'AB' and 'BA' flow directions of the calibration curve fit and for the corrected data on which it is based. Examples of program output are shown in figure 2 from the thermal printer ('AB' direction only) and in figure 3 from the plotter.

DATE (DAY MONTH YR) 31/7/75
METER S/N 6014
METER RANGE 8.33 CCPS (.5 LPM)

FLOW DIRECTION IS AB

| NO | LEFT | RGHT | P2 | VOL | TIME | T2 | PO | BARP | DIFF |
|---|---|---|---|---|---|---|---|---|---|
| 1 | 0.00 | 0.00 | 0.00 | 0.0 | 1.0000 | 0.0 | 0.0 | 760.0 | 0.00 |
| 2 | 3.29 | 4.87 | 1.50 | 0.2 | 1.3840 | 22.1 | 1.5 | 758.0 | -1.58 |
| 3 | 5.24 | 6.90 | 1.50 | 0.2 | 0.9220 | 22.1 | 1.5 | 758.0 | -1.66 |
| 4 | 8.00 | 9.62 | 1.60 | 0.4 | 1.2691 | 22.1 | 1.5 | 758.0 | -1.62 |
| 5 | 11.02 | 12.71 | 1.70 | 0.4 | 0.9375 | 22.1 | 1.5 | 758.0 | -1.69 |
| 6 | 12.81 | 14.58 | 1.80 | 0.6 | 1.2150 | 22.1 | 1.5 | 757.9 | -1.77 |
| 7 | 14.51 | 16.30 | 1.90 | 0.6 | 1.0770 | 22.1 | 1.5 | 757.9 | -1.79 |
| 8 | 17.58 | 19.36 | 2.00 | 0.8 | 1.1919 | 22.1 | 1.5 | 757.9 | -1.78 |
| 9 | 19.58 | 21.39 | 2.10 | 0.8 | 1.0713 | 22.1 | 1.5 | 757.9 | -1.81 |
| 10 | 22.72 | 24.57 | 2.20 | 1.0 | 1.1540 | 22.0 | 1.5 | 757.9 | -1.85 |
| 11 | 25.78 | 27.47 | 2.30 | 1.0 | 1.0190 | 22.0 | 1.5 | 758.0 | -1.69 |
| 12 | 6.11 | 7.81 | 63.00 | 0.4 | 1.0188 | 22.1 | 1.5 | 758.0 | -1.70 |
| 13 | 11.65 | 13.47 | 67.70 | 0.6 | 0.8207 | 22.1 | 1.5 | 758.0 | -1.82 |

NO. DATA SETS IS 13

COEF'S FOR 'DELTA P=A+B*FLOW+C*(FLOW*FLOW)'
OR 'FLOW=(-B+SQR<B↑2-4*C*(A-DELTA P)>)/(2*C)'

WHERE DELTA P IS IN MILLIBARS, FLOW IS IN C.C./SEC

ZERO COEF = A =-6.9785E-03
FIRST DEG COEF = B = 3.3637E+00
SECOND DEG COEF= C =-3.0688E-03

| DELP(DATA) | FLOW(DATA) | FLOW(CALC) | DEVIATION | % DVATION | % FULL RANGE |
|---|---|---|---|---|---|
| 0.0000E+00 | 1.6666E-08 | 2.0746E-03 | -2.0746E-03 | -1.2447E+07 | -2.4895E-02 |
| 7.9506E+00 | 2.3726E+00 | 2.3708E+00 | 1.8592E-03 | 7.8360E-02 | 2.2310E-02 |
| 1.1828E+01 | 3.5548E+00 | 3.5298E+00 | 2.4930E-02 | 7.0132E-01 | 2.9916E-01 |
| 1.7168E+01 | 5.1467E+00 | 5.1298E+00 | 1.6833E-02 | 3.2708E-01 | 2.0200E-01 |
| 2.3121E+01 | 6.9403E+00 | 6.9194E+00 | 2.0892E-02 | 3.0102E-01 | 2.5070E-01 |
| 2.6687E+01 | 8.0111E+00 | 7.9942E+00 | 1.6919E-02 | 2.1120E-01 | 2.0303E-01 |
| 3.0020E+01 | 9.0144E+00 | 9.0005E+00 | 1.3930E-02 | 1.5453E-01 | 1.6716E-01 |
| 3.5993E+01 | 1.0819E+01 | 1.0808E+01 | 1.0133E-02 | 9.3659E-02 | 1.2159E-01 |
| 3.9919E+01 | 1.2002E+01 | 1.2001E+01 | 1.7258E-03 | 1.4378E-02 | 2.0710E-02 |
| 4.6090E+01 | 1.3874E+01 | 1.3879E+01 | -5.6695E-03 | -4.0864E-02 | -6.8034E-02 |
| 5.1899E+01 | 1.5654E+01 | 1.5654E+01 | -5.4929E-04 | -3.5089E-03 | -6.5915E-03 |
| 1.3573E+01 | 4.0246E+00 | 4.0522E+00 | -2.7671E-02 | -6.8756E-01 | -3.3206E-01 |
| 2.4496E+01 | 7.2623E+00 | 7.3335E+00 | -7.1266E-02 | -9.8132E-01 | -8.5520E-01 |

SUM OF SQRS OF DEVIATION= 7.8122E-03

| DELP(CALC) | FLOW(CALC) |
|---|---|
| 0.0000E+00 | 2.0746E-03 |
| 5.0000E+00 | 1.4905E+00 |
| 1.0000E+01 | 2.9830E+00 |
| 1.5000E+01 | 4.4796E+00 |
| 2.0000E+01 | 5.9804E+00 |
| 2.5000E+01 | 7.4853E+00 |
| 3.0000E+01 | 8.9944E+00 |
| 3.5000E+01 | 1.0507E+01 |
| 4.0000E+01 | 1.2025E+01 |
| 4.5000E+01 | 1.3547E+01 |
| 5.0000E+01 | 1.5073E+01 |

Figure 2. Program output from thermal printer for 'AB' direction.




Figure 3. Program output from plotter.

## 4. USE OF PROGRAM OUTPUT

The flow meter calibration coefficients calculated by this program provide a relationship between the volume flow rate at the midpoint of the meter and the measured pressure drop across its signal ports. The relationship can be expressed as equation (12a):

$$\Delta P_s = A + BQ_m + CQ_m^2 \quad ,$$

by which it is assumed that the gas is air and that it is at standard conditions (20°C, 1 bar) at the midpoint of the meter. If standard conditions do not exist at the meter midpoint, then $\Delta P_S$ must be calculated by use of equation (11a) before $Q_m$ can be found from either equation (12a) or from the program's plot output.

If the flow of a gas other than air is measured, the values of B and C must be modified to reflect the resulting changes in $\mu$ and $\rho$, since from equation (12),

$$B = b\mu_s \text{ and } C = c\rho_s \quad .$$

Therefore, equation (12a) is modified to become

$$\Delta P_s = A + B_1 Q_m + C_1 Q_m^2 \quad , \tag{14}$$

where

$$B_1 = B \cdot \frac{\mu_{s_g}}{\mu_{s_{AIR}}} = B \cdot \frac{\mu_{s_g}}{181.5 \times 10^{-7}} \quad ,$$

$\mu_{s_g}$ is the viscosity of the gas being used (as given in table I),

$C = C \cdot (S.G.)$ ,

S.G. is the specific gravity of the gas being used (as given in table I).

Finally, if the equivalent volume flow at conditions other than those that exist at the midpoint of the meter (e.g., standard conditions) is desired, then equation (13) can be used.

```
10 REM *****CALFLO**20 MAR 75**M.FUNKE*********************
15 REM *****USE TO CALIBRATE VOL-O-FLO METERS**************
20 REM *****EQ FOR FIT IS LEAST SQUARES AND SECOND ORDER***
30 DIM P(31),Q(31),L(31),R(31),A(2,3),U(3),T(2),B(2,3),C(3),K(3)
40 SELECT PRINT 005(64)
50 PRINT "INSTRUCTIONS:"
60 PRINT "  1. MAX NO. DATA SETS ENTERED IS 30."
70 PRINT "  2. DATA SETS SHOULD BEGIN WITH 'AB' DIRECTION DATA,
              TERMINATED BY 'LEFT' = 9.999, FOLLOWED BY 'BA'
              DATA, TERMINATED BY 'LEFT' = 9.999."
80 PRINT "  3. DATA CORRECTIONS ARE ALLOWED AFTER ALL DATA FOR
              CURRENT DIRECTION HAVE BEEN ENTERED."
90 PRINT "  4. ON PLOTTER, SET ZERO REFERENCE IN LOWER LEFT
              CORNER; SET SCALE AT X=Y=+8 INCHES."
100 STOP "                KEY CONTINUE, EXECUTE"
110 PRINT HEX(03)
120 INPUT "DATE (DAY MONTH YR)      ", D$
130 INPUT "METER S/N                ", S
140 INPUT "METER RANGE(LPM)         ", R$:CONVERT R$TO R:R=
    R*1000/60
150 D1$="AB"
160 SELECT PRINT 015(80)
170 PRINT :PRINT
180 PRINT "DATE (DAY MONTH YR)       "; D$
190 PRINT "METER S/N                "; S
200 PRINT "METER RANGE              ";
210 IF R[1THEN 240
220 IF R[10THEN 250
230 IF R[100THEN 260:GOSUB 300:GOTO 310
240 GOSUB 270:GOTO 310
250 GOSUB 280:GOTO 310
260 GOSUB 290:GOTO 310
270 PRINTUSING 1840,R;:RETURN
280 PRINTUSING 1860,R;:RETURN
290 PRINTUSING 1880,R;:RETURN
300 PRINTUSING 1900,R;:RETURN
310 PRINT " CCPS (";R$;:PRINT " LPM)"
320 REM *****NORMALIZE AND PRINT DATA***
330 FOR K=1TO 3:K(K)=0:NEXT K
340 L=0: M=2: PRINT :PRINT :PRINT :PRINT "FLOW DIRECTION IS ";
    D1$:PRINT
350 PRINTUSING 360,"NO","LEFT","RGHT","P2","VOL","TIME","T2"
 ,"PO","BARP","DIFF"
360 % ##    ####    ####    ##    ###    ####    ####  ##   ####
   ####
```

```
370 L(1),R(1),B,T1,P0=0:V=1E-9:I,T=1:B1=760:I=1:N3=0
380 D=L(I)-R(I):PRINTUSING 400,I,L(I),R(I),B,V,T,T1,P0,B1,D
390 SELECT PRINT 005:PRINT HEX(03):PRINT "DATA POINT";I+1:SELECT
    PRINT 015(80)
400 % ## -##.## -##.##  -##.## -##.# -#.#### -##.# -#.# -###.#
 -##.##
410 P(I)=(L(I)+R(I))*2.4884/2.54
420 Q(I)=1000*V/T/60*(1.8717*P0+B1)/(.75218*P(I)/2+7.5218*B+B1)
430 P(I)=P(I)/(1+9/5*.00146*(T1-20)+.0013/760*(752.18-.75218*P(I
   )/2-7.5218*B-B1)):I=I+1:IF N2=1THEN 550:N9=I
440 INPUT "LEFT MANOMETER READING(CM.W.C.)=",L(I)
450 IF L(I)=9.999THEN 550
460 INPUT "RIGHT MANOMETER READING(CM.W.C.)=",R(I)
470 INPUT "BACK PRESSURE(KPA)=",B
480 INPUT "VOLUME(LITERS)=",V
490 INPUT "TIME(MIN)=",T
500 INPUT "TEMPERATURE(DEG C)=",T1
510 INPUT "VOLUME PRESSURE(IN.W.C.)=",P0
520 INPUT "BAROMETRIC PRESSURE(MM.HG.)=",B1:GOTO 380
530 REM *****DATA CORRECTIONS***
540 N8=0
550 N2=0:INPUT "DO YOU WANT TO MAKE DATA CORRECTIONS(1=YES)",
 N2
560 IF N2=1THEN 570:IF N3=1THEN 880:GOTO 1120
570 IF N8=1THEN 580:PRINT :PRINT "POINTS CORRECTED INCLUDE"
580 N8=1:INPUT "WHICH DATA POINT NUMBER",I:GOTO 440
590 REM *****PRINT COEFF'S***
600 SELECT PRINT 015(80)
610 PRINT :PRINT
620 Z=0:FOR K=1TO 3
630 IF K(K)[1THEN 640:IF K(K)]I-1THEN 640:IF K(K)[]INT(K(K))THEN
    640:GOTO 650
640 K(K)=0
650 IF K(K)=0THEN 670:IF Z=1THEN 660:PRINT "POINTS DELETED INCLU
DE":Z=1
660 PRINT K(K)
670 NEXT K
680 PRINT "NO. DATA SETS IS "; I-1-Z1
690 PRINT :PRINT "  COEF'S FOR 'DELTA P=A+B*FLOW+C*(FLOW*FLOW)'"
    :PRINT "          OR  'FLOW=(-B+SQR[B!2-4*C*(A-DELTA P)])/(2*
C)'"
700 PRINT :PRINT "       WHERE    DELTA P IS IN MILLIBARS,";:
    PRINT " FLOW IS IN C.C./SEC"
710 PRINT
720 PRINT "ZERO COEF      = A =";:PRINTUSING 730, C(1):PRINT
    "FIRST DEG COEF = B =";:PRINTUSING 730, C(2): PRINT
    "SECOND DEG COEF= C =";: PRINTUSING 730, C(3)
```

```
730 %-#.####!!!!
740 REM *****PRINT DATA ANALYSIS***
750 PRINT :PRINTUSING 760,"DELP(DATA)","FLOW(DATA)","FLOW(CALC)"
    ,"DEVIATION","% DVATION","% FULL RANGE"
760 % ##########   ##########   ##########   #########    ######
###    ############
770 S1,Z=0:FOR N=1TO I-1:FOR K=1TO 3:IF K(K)[]NTHEN 780:Z=1
780 NEXT K:IF Z=1THEN 830
790 X1=(-C(2)+SQR(C(2)*C(2)-4*C(3)*(C(1)-P(N))))/(2*C(3))
800 W=Q(N)-X1: S1=S1+W*W: WO=100*W/Q(N): W5=100*W/R
810 PRINTUSING 820, P(N),Q(N),X1,W,WO,W5
820 %-#.####!!!!  -#.####!!!!  -#.####!!!!  -#.####!!!!  -#.####
!!!!  -#.####!!!!
830 Z=0:NEXT N: PRINT
840 PRINT "    SUM OF SQRS OF DEVIATION=";:PRINTUSING 850, S1
850 %-#.####!!!!
860 N3=1:GOTO 530
870 REM *****DATA DELETIONS***
880 K=0
890 N4=0:INPUT "DO YOU WANT TO DELETE A DATA POINT(1=YES)",N4
900 IF N4=1THEN 910:IF N3=2THEN 1120:IF N8=1THEN 1120:GOTO 930
910 K=K+1:N3=2
920 INPUT "WHICH DATA POINT NUMBER",K(K):IF K[3THEN 890:GOTO 112
0
930 IF D1$="BA"THEN 970
940 REM *****SET PRESSURE AND FLOW SCALE***
950 04=2*R
960 F1=625/50:F2=625/04
970 REM *****GENERATE PRESS-FLOW TABLE***
980 PRINT : PRINTUSING 990, "DELP(CALC)","FLOW(CALC)"
990 % ##########    ##########
1000 IF D1$="BA"THEN 1010:GOSUB '18:GOTO 1020
1010 GOSUB '19
1020 FOR N=0 TO 10: Y=5*N
1030 X1=(-C(2)+SQR(C(2)*C(2)-4*C(3)*(C(1)-Y)))/(2*C(3))
1040 PRINTUSING 1050, Y,X1
1050 %-#.####!!!!   -#.####!!!!
1060 REM *****PLOT PRESS-FLOW CURVE***
1070 IF N]0THEN 1080:GOSUB '11(Y,X1):GOTO 1090
1080 GOSUB '12(Y,X1)
1090 NEXT N
1100 SELECT PRINT 005(64):GOTO 1280
1110 REM *****COEFF CALCULATION***
1120 I0,I1,I2,K0,K1,K2,K3,K4,K5,K6,K7=0:W=1:I=N9:Z1,Z=0
1130 FOR N=1TO I-1:FOR K=1TO 3:IF K(K)[]NTHEN 1140:Z=1:Z1=Z1+1
1140 NEXT K:IF Z=1THEN 1160
1150 I0=I0+W*Q(N):K0=K0+P(N)*W:K4=K4+W
```

```
1160 Z=0:NEXT N:K0=K0/K4:I0=I0/K4
1170 FOR N=1TO I-1:FOR K=1TO 3:IF K(K)[]NTHEN 1180:Z=1
1180 NEXT K:IF Z=1THEN 1200
1190 H1=Q(N)-I0:I1=I1+Q(N)*H1*H1*W:K1=K1+P(N)*H1*W:K5=K5+H1*H1*W
1200 Z=0:NEXT N:K1=K1/K5:J1=K5/K4:I1=I1/K5
1210 FOR N=1TO I-1:FOR K=1TO 3:IF K(K)[]NTHEN 1220:Z=1
1220 NEXT K:IF Z=1THEN 1240
1230 H2=(Q(N)-I0)*(Q(N)-I1)-J1:I2=I2+Q(N)*H2*H2*W:K2=K2+P(N)*H2
     *W:K6=K6+H2*H2*W
1240 Z=0:NEXT N:K2=K2/K6:J2=K6/K5:I2=I2/K6
1250 C(1)=K0-K1*I0+K2*(I0*I1-J1)
1260 C(2)=K1-K2*(I0+I1):C(3)=K2:GOTO 590
1270 REM *****SUBROUTINE FOR PLOTTING***
1280 V3=5: V4=04/5
1290 GOSUB '20(V3,V4)
1300 V4=50: V5=10: V7=04: V8=04/5
1310 GOSUB '21(V4,V5,V7,V8):PLOT [,,R]:GOTO 2000
1320 REM *****ROUNDOFF COMPENSATION***
1330 DEFFN'10(X9,Y9):X8=INT(F1*X9-X0+.5):Y8=INT(F2*Y9-Y0+.5):
     X0=X0+X8:Y0=Y0+Y8:RETURN
1340 REM *****PLOT POINT UP***
1350 DEFFN'11(X9,Y9):GOSUB '10(X9,Y9):PLOT [X8,Y8,U]:RETURN
1360 REM *****PLOT POINT DOWN***
1370 DEFFN'12(X9,Y9):GOSUB '10(X9,Y9):PLOT [X8,Y8,D]:RETURN
1380 REM *****PLOT SYMBOL***
1390 DEFFN'17(X9,Y9):GOSUB '10(X9,Y9):PLOT [X8,Y8,F$]:RETURN
1400 REM *****GOTO ORIGIN OF 'AB' PLOT***
1410 DEFFN'18:X0,Y0=0:PLOT [,,R],[125,125,U]: RETURN
1420 REM *****GOTO ORIGIN OF 'BA' PLOT***
1430 DEFFN'19:X0,Y0=0:PLOT [,,R],[500,62,U],[500,63,U]: RETURN
1440 REM *****DRAW AXIS***
1450 DEFFN'20(V3,V4)
1460 IF D1$="AB"THEN 1480
1470 GOSUB '19:GOTO 1490
1480 GOSUB '18
1490 V5=0
1500 GOSUB '12(V5,0)
1510 PLOT [,10,D],[,-20,D],[,10,D]
1520 V5=V5+V3
1530 IF V5[=50 THEN 1500
1540 V5=0
1550 GOSUB '12(0,V5)
1560 PLOT [10,,D],[-20,,D],[10,,D]
1570 V5=V5+V4
1580 IF V5[=04 THEN 1550
1590 RETURN
1600 REM *****SCALE AXIS***
```

```
1610 DEFFN'21(V4,V5,V7,V8)
1620 IF D1$="AB"THEN 1640
1630 GOSUB '19:GOTO 1650
1640 GOSUB '18
1650 SELECT PRINT 413
1660 GOSUB 1940
1670 V9=0
1680 GOSUB '10(V9,0)
1690 X8=X8-39: X0=X0+28: Y8=Y8-20: Y0=Y0-20
1700 PLOT [X8,Y8,U]
1710 O=V9: GOSUB 1970
1720 V9=V9+V5
1730 IF V9[=V4 THEN 1680: IF D1$="AB"THEN 1740: GOSUB '19:GOTO
     1750
1740 GOSUB '18
1750 V9=0
1760 PLOT [-120,0,U]
1770 O=V9:IF O=0THEN 1810:IF O[1THEN 1800
1780 IF O[10THEN 1810
1790 IF O[100THEN 1820:GOSUB 1890:GOTO 1910
1800 GOSUB 1830:GOTO 1910
1810 GOSUB 1850:GOTO 1910
1820 GOSUB 1870:GOTO 1910
1830 PRINTUSING 1840,O;:RETURN
1840 %-.###
1850 PRINTUSING 1860,O;:RETURN
1860 %-#.##
1870 PRINTUSING 1880,O;:RETURN
1880 %-##.#
1890 PRINTUSING 1900,O;:RETURN
1900 %-###.
1910 PLOT [-110,125,U]:V9=V9+V8
1920 IF V9[=V7 THEN 1770
1930 SELECT PRINT 005:RETURN
1940 PLOT [2,,C],[22,,S]: RETURN
1950 PRINTUSING 1960,O;: RETURN
1960 %-###
1970 PRINTUSING 1980,O;:RETURN
1980 %-##
1990 REM *****LABEL AXIS***
2000 PLOT [2,,C],[25,,S],[,,R]
2010 SELECT PRINT 413
2020 PLOT [225,40,U]:IF D1$="AB"THEN 2030:PLOT 2[438,0,U]
2030 PRINT "DELTA P (MILLIBARS)"
2040 PLOT [,,R],[100,775,U]:IF D1$="AB"THEN 2050:PLOT 2[438,0,U]
2050 PRINT "FLOW (CC/SEC)":IF D1$="AB"THEN 2060:PLOT [390,361
     ,U]:PRINT ";";I-1-Z1;:PLOT [-20,0,U]:PRINT "'BA'":PLOT [-85
0,   -185,U]:PRINT " 'BA' DIRECTION":GOTO 2260
```

```
2060 REM *****LABEL PLOT***
2070 PLOT [-500,0,U],[60,450,U]
2080 PRINT "METER S/N"
2090 PLOT [-225,-45,U]
2100 PRINT "METER RANGE"
2110 PLOT [-275,-45,U]
2120 PRINT "NO. OF DATA POINTS"
2130 PLOT [-450,-45,U]
2140 PRINT "DATE"
2150 PLOT [450,-48,U]
2160 PRINT "COEFF'S FOR P=A+BQ+CQ";:PLOT [-35,10,U],[1,,C]:PRINT
     2:PLOT [-650,-50,U],[2,,C]:PRINT "OR  Q=(-B+SQR(B";:PLOT
     [-30,10,U],[1,,C]:PRINT 2:PLOT [-30,-10,U],[2,,C]:PRINT
     "-4C(A-P)))/2C"
2170 PLOT [275,225,U]:PRINT S:PLOT [-150,-45,U]
2180 IF R[1THEN 2210
2190 IF R[10THEN 2220
2200 IF R[100THEN 2230:GOSUB 300:GOTO 2240
2210 GOSUB 270:GOTO 2240
2220 GOSUB 280:GOTO 2240
2230 GOSUB 290
2240 PRINT " CCPS":PLOT [-250,-45,U]:PRINT I-1-Z1;:PLOT [-20,0,U
]:   PRINT "'AB'":PLOT [-160,-45,U]:PRINT D$:PLOT 3[-505,-46,U]
2250 PRINT " 'AB' DIRECTION"
2260 PLOT [-275,-50,U]
2270 PRINT "A =";:PRINTUSING 730,C(1)
2280 PLOT 2[-175,-20,U]
2290 PLOT [2,,C],[25,,S]
2300 PRINT "B =";:PRINTUSING 730,C(2)
2310 PLOT 2[-175,-20,U]
2320 PLOT [2,,C],[25,,S]
2330 PRINT "C =";:PRINTUSING 730,C(3):PLOT [,,R]:SELECT PRINT
     005
2340 F$="+":PLOT [1,,C],[0,,S]:IF D1$="BA"THEN 2350:GOSUB '18:
     GOTO 2360
2350 GOSUB '19
2360 FOR N=1TO I-1:FOR K=1TO 3:IF K(K)[]NTHEN 2370:Z=1
2370 NEXT K:IF Z=1THEN 2390
2380 GOSUB '17(P(N),Q(N))
2390 Z=0:NEXT N:IF D1$="BA"THEN 2400:D1$="BA":SELECT PRINT 015(
     80):GOTO 320
2400 SELECT PRINT 005:PLOT [,,R]:END
```

## DISTRIBUTION

DEFENSE DOCUMENTATION CENTER
CAMERON STATION, BUILDING 5
ALEXANDRIA, VA 22314
ATTN DDC-TCA (12 COPIES)

OFC, CHIEF OF RESEARCH & DEVELOPMENT
USA RSCH & DEV GROUP (EUROPE)
BOX 15
FPR NEW YORK 09510
ATTN LTC EDWARD E. CHICK
CHIEF, MATERIALS BRANCH

COMMANDER
US ARMY MATERIEL DEVELOPMENT
& READINESS COMMAND
5001 EISENHOWER AVENUE
ALEXANDRIA, VA 22333
ATTN DRCRD, RES, DEV,
& ENGR DIRECTORATE

COMMANDER
USA ARMAMENT COMMAND
ROCK ISLAND, IL 61201
ATTN DRSAR-ASF, FUZE DIV
ATTN DRSAR-RDF, SYS DEV DIV - FUZES

COMMANDER
USA MISSILE & MUNITIONS
CENTER & SCHOOL
REDSTONE ARSENAL, AL 35809
ATTN ATSK-CTD-F

COMMANDER
PICATINNY ARSENAL
NUCLEAR DEVELOPMENT & ENGR. DIR
DOVER, NJ 07801
ATTN SARPA, MR. A. FRANZ
ATTN SARPA-ND-C-C, MR. DAVID SAMPAR

COMMANDER
NAVAL AIR SYSTEMS COMMAND
WASHINGTON, DC 07801
ATTN AIR 52022A1,
MR. WILLIAM LAUGHREY

NASA, LANGLEY STATION
HAMPTON, VA 23365
ATTN MS 494, MR. H. DOUGLAS GARNER

COMMANDER
AIR FORCE FLIGHT DYNAMICS LAB
FLIGHT CONTROL DIVISION (FDCL)
WRIGHT-PATTERSON AIR FORCE BASE, OH 45433
ATTN MR. HARRY M. SNOWBALL

COMMANDER
US NAVAL AIR DEVELOPMENT CENTER
AEROMECHANICS DEPARTMENT
WARMINSTER, PA 18974
ATTN MR. HORACE WELK

COMMANDER
US ARMY MISSILE COMMAND
REDSTONE ARSENAL, AL 35809
ATTN DRSMI-RGC, MR. JAMES WILLIAMS

COMMANDER
NAVAL AIR SYSTEMS COMMAND
DEPARTMENT OF THE NAVY
WASHINGTON, DC 20360
ATTN CODE AIR 52022A, MR. JOHN BURNS

COMMANDER
NAVAL SHIP SYSTEMS COMMAND
WASHINGTON, DC 20360
ATTN SHIPS 03414, MR. ARTHUR CHAIKIN

AIR FORCE INSTITUTE OF TECHNOLOGY
WRIGHT-PATTERSON AIR FORCE BASE, OH 45433
ATTN AFIT/ENM, MR. M. E. FRANKE

COMMANDER
US ARMY WEAPONS COMMAND
ARMORED WEAPONS SYSTEMS DIR
ROCK ISLAND, IL 61201
ATTN SWERR-T-AC, LORRAINE D. WRIGHT

COMMANDER
USAMERDC
FT. BELVOIR, VA 22060
ATTN SMEFB-EP, MR. ROBERT WARE

HARRY DIAMOND LABORATORIES
ATTN MCGREGOR, THOMAS, COL, COMMANDING
OFFICER/FLYER, I.N./LANDIS, P.E./
SOMMER, H./CONRAD, E.E.
ATTN CARTER, W.W., DR., ACTING TECHNICAL
DIRECTOR/MARCUS, S.M.
ATTN KIMMEL, S., IO
ATTN CHIEF, 0021
ATTN CHIEF, 0022
ATTN CHIEF, LAB 100
ATTN CHIEF, LAB 200
ATTN CHIEF, LAB 300
ATTN CHIEF, LAB 400
ATTN CHIEF, LAB 500
ATTN CHIEF, LAB 600
ATTN CHIEF, DIV 700
ATTN CHIEF, DIV 800
ATTN CHIEF, LAB 900
ATTN CHIEF, LAB 1000
ATTN RECORD COPY, BR 041
ATTN HDL LIBRARY (3 COPIES)
ATTN CHAIRMAN, EDITORIAL COMMITTEE
ATTN CHIEF, 047
ATTN TECH REPORTS, 013
ATTN PATENT LAW BRANCH, 071
ATTN MCLAUGHLIN, P.W., 741
ATTN CHIEF, 310 (15 COPIES)
ATTN CHIEF, 340 (10 COPIES)
ATTN CAMPAGNUOLO, C., 940
ATTN FUNKE, M. F. (5 COPIES)

**DEPARTMENT OF THE ARMY**
HARRY DIAMOND LABORATORIES
2800 POWDER MILL RD
ADELPHI, MD 20783

OFFICIAL BUSINESS
PENALTY FOR PRIVATE USE $300

AN EQUAL OPPORTUNITY EMPLOYER

POSTAGE AND FEES PAID
DEPARTMENT OF THE ARMY
DOD 314

**THIRD CLASS**